
Subject: Re: stippling or cross hatching in contour plot
Posted by [David Fanning](#) on Mon, 08 Oct 2007 16:47:28 GMT
[View Forum Message](#) <> [Reply to Message](#)

Whoops! Sorry. trying to do too many things at once.

Try this code.

```
;; create some data
array=dist(10,10)

Window,2
loadct,2
;; contour data with color
loadct, 33, ncolors=5, bottom=1
device, decomposed=0, get_decomposed=theState

thisDevice = !D.Name
xsize = !D.X_Size
ysize = !D.Y_Size
Set_Plot, 'Z'
Device, Set_Resolution=[xsize,ysize], Z_Buffer=0
contour,array,levels=indgen(5),c_colors=indgen(5)+1,/fill, $
  xstyle=1, ystyle=1, position=[0.1, 0.1, 0.9, 0.9]
;; try to put hatching on top
locx = [0.1, 0.9] * xsize
locy = [0.1, 0.9] * ysize
snap1 = TVRD(locx[0], locy[0], locx[1]-locx[0]+1, $
  locy[1]-locy[0]+1)

contour, array,levels=indgen(5),/overplot, $
  c_orientation=[30,45,-30,-45,0]
snap2 = TVRD(locx[0], locy[0], locx[1]-locx[0]+1, $
  locy[1]-locy[0]+1)

Set_Plot, thisDevice
contour,array, xstyle=1, ystyle=1, /NoData, $
  position=[0.1, 0.1, 0.9, 0.9], levels=indgen(5)
TV, snap1 > snap2, locx[0], locy[0]
device, decomposed=theState
end

--
David Fanning, Ph.D.
Fanning Software Consulting, Inc.
Coyote's Guide to IDL Programming: http://www.dfanning.com/
```
